# Molecular Coding Format manual

Author : Akira Yamaji  Date : May 5, 2016
Located at : http://www.ctan.org/pkg/mcf2graph

Molecular Coding Format(MCF) is new linear notation represent chemical structure diagrams. This 'Coding' is named from coding(programing) technique like adressing,grouping,macro,etc. There are no Meta language commands in MCF. mcf2graph.mf(metapost/metafont macro) convert MCF file to graphics file pk font,PNG,SVG,EPS or MDL MOL file(V2000).

**(Molecular definition file)**

```
%-------------------------------------------------------------------------
input mcf2graph.mf;                                 % input macro
%-------------------------------------------------------------------------
sw_auxout:=1;          % aux(information) file output on > Gloval setting
sw_numberA:=0;         % numbering atom off              >
sw_numberB:=0;         % numbering bond off              >
sw_expand:=0;          % substituent expand mode off     >
sw_start_vector:=0;    % start vector output off         >
sw_font_frame:=0;      % font frame off                  >
font_wd#:=60mm#;       % font width                      >
font_ht#:=40mm#;       % font height                     >
var3:="cal_MW"; tag3:="cMW";                            > AUX file table
var4:="cal_FM"; tag4:="cFM";                            >
%%%% var5:="len_x";   tag5:="Lx";                       >
%%%% var6:="len_y";   tag6:="Ly";                       >
%%%% sw_auxfix:=1;    % fixed aux file                  > fixed AUX file
%%%% auxtag_out;      % tag list output                 >
outputformat:="png"; hppp:=vppp:=0.1;                   > PNG output
%%%% outputformat:="svg";                               > SVG output
outputtemplate:="%j-%3c.png";                           >
%-------------------------------------------------------------------------
beginfont("NO:1","EN:Chlorophyll a","MW:893.49")  % begin font(information)
  sw_font_frame:=1;     % font frame on                 >
  font_wd#:=120mm#;     % font width set                >  Local setting
  font_ht#:=80mm#;      % font height set               >
  %-----------------------------------------------------------------------
  MCf(                                                  % begin MCF
    <54,{,‘‘1,?5,$(2,5)d,4:N,3\,54~dl,                  %
    |,?5,$(2,4)d,5:N,                                  %
   -2\,54~dl,|,?5,2=d,5:N,-2\~dl,54,                    %
   |,?5,5=d,5:N,-2\~dl,&#5,                            %
   -1*,24,/*COOMe^15,72,//O,&#1,},                      %
   4\‘1.48,Mg,&17,-1*,&11~vb,-1*,&23~vb,                %
   @(2,9,15,20~zf)/Me,8:/Et,14\,!~dr,                   %
   21*,-6~wf,!2,//O,!,O,!2,!~dl,                        %
   |,!13,@(1,5,9,13)/Me,                               %
  )                                                     % end MCF
endfont                                                 % end font
%-------------------------------------------------------------------------
bye
```

**(Molecular structure diagram output)**



**(Molecular information output)**
'filename'-info.aux : for use in TEX, it takes over filename, char number, molecular information,etc.

```
(sw_auxfix=0)
F:mcf_man_soc;C:0;cMW:893.48962;cFM:C55H72MgN4O5;NO:1;EN:Chlorophyll a;MW:893.49

(sw_auxfix=1)
F;C;cMW;cFM;NO;EN;MW
mcf_man_soc;0;893.48962;C55H72MgN4O5;1;Chlorophyll a;893.49
```

( tag : variable )
F:filename C:char number,cMW:molecular weight calculated, cFM:molecular formula calculated
NO:serial number, EN:english name, MW:molecular weight from literature data

**(LaTeX file example)**

```
%-------------------------------------------------------------------------
\documentclass[a4paper]{article}
\usepackage{graphicx}
\pagestyle{empty}
\makeatletter%
%-------------------------------------------------------------------------
\def\@F{F}\def\@C{C}\def\@EN{EN}\def\@NO{NO}\def\@MW{MW}\def\@FMc{FMc}%
\def\@fst@param#1:#2;{#1}\def\@sec@param#1:#2;{#2}%
\def\mol@sel#1{%
\if#1\empty\relax\else%
  \edef\@tag{\expandafter\@fst@param#1;}%
  \edef\@var{\expandafter\@sec@param#1;}%
  \ifx\@tag\@F\edef\MOLfile{\@var}\fi%
  \ifx\@tag\@C\edef\MOLchar{\@var}\fi%
  \ifx\@tag\@EN\edef\MOLnameE{\@var}\fi%
  \ifx\@tag\@NO\edef\MOLnum{\@var}\fi
  \ifx\@tag\@MW\edef\MOLmw{\@var}\fi
  \ifx\@tag\@FMc\edef\CALfm{\@var}\fi
\fi}%
\def\put@char{%
  \begin{picture}(120,45)%
     \put(10,35){\bf [\MOLnum] \MOLnameE}%
     \put(16,30){\sf MW:\MOLmw}%
     \put(16,25){\sf FM:\CALfm}%
     \put(0,0){\font\@strufont=\MOLfile\relax%
               \hbox{\@strufont\char\MOLchar}}%
  \end{picture}}%
\def\INFO#1{\@for\@temp:=#1\do{\mol@sel\@temp}\put@char}%
\makeatother
%-------------------------------------------------------------------------
\begin{document}
\unitlength=1mm%
\INFO{F:mcf_man_soc ,C:0 ,cMW:893.48962 ,cFM:C55H72MgN4O5%
      ,NO:1 ,EN:Chlorophyll a ,MW:893.49}%
\end{document}
%-------------------------------------------------------------------------
```

**[1 ] Chlorophyll a**
MW:893.49
FM:C55H72MgN4O5



## No.1 Chain(1)

```
plus (+): anticlockwize
minus(-): clockwize

  <30,-60,60,-90,120,-90,60
```




## No.2 Chain(2)

```
!  : take value 60 or -60 depend on
        current angle and enviroment
!6 : !,!,!,!,!,!

  <30,!,!,!,!,!,!
  <30,!6
```




## No.3 Jump to atom and branch bond

```
3* : Jump to A3
3\ : 3*,0

  <30,!6,3*,0,!
  <30,!6,3\,  !

** An(-999<=n<=4095): atom number
```




## No.4 Connect atom

```
&5 : Connect to A5

  <30,!6,3\,!.&5
```




## No.5 Ring

```
?n : n membered ring

  ?6

  ** ?6 : <-120,60,60,60,60,60,&1
  ** ?n(3<=n<=20)
  ** bold arrow is default angle and position
```



No.6 **Rotate current angle**

```
<angle : rotate current angle

  <30,!4,<30,?6
```



No.7 **Change bond type (1)**

```
a~type : ~~type,a
dm:double middle,
dl:double left side, dr:double right side,
wf:wedge forward, wb:wedge backward,
zf:wedge dotted, zb:wedge dotted backward

  <30,!~dm,!~dl,!~dr,!~wf,!~wb,!~zf,!~zb
```



No.8 **Change bond type (2)**

```
Bn=bond type : change bond type at Bn
vf : vector forward
vb : vector backward
tm : triple
dt : dotted
wv : wave

  <30,!6,1=vf,2=vb,3=tm,4=dt,5=bz,5=wv
```



No.9 **Change bond type (3)**

```
bd : broad single
bz : broad dotted
ov : over line
$(2,4)dr : 2=dr,4=dr

  <30,!5,1=bz,$(2,4)dr,
        @(3~bd^-45‘2,5~ov^45‘2)/Me

** 3~bd^-45‘2 : ~~bd,^^-45,’‘2,3
```



No.10 **Change bond length (1)**

```
(!,!n,?n)‘length : change length of !,!n,?n

  <30,!2,!2‘1.2,!2

** !2‘1.2 : ’‘1.2,!2
```



No.11 **Change bond length (2)**

```
‘‘length : change all bond length after

  <30,!2,‘‘1.2,!4
```




No.12 **Change ring length**

```
?n‘length : change ring length

  <30,?6,3\,?6‘1.2,11=?6
```



No.13 **Change atom (1)**

```
Insert hetero atom

  <30,!2,O,!2
```




No.14 **Change atom (2)**

```
2:O : change A2 C to O
@(3,5)N : change  A3,A5 C to N

  <30,!6,2:O,@(3,5)N
```




No.15 **Change atom (3)**

```
2:N : change A2 C to N

  ?6,4\,?6,2:N
```



No.16 **Change atom (brock address)**

```
|  : divide brock
|| : reset brock adress

?6,4\,|,?6,2:O,||,2:N
```




No.17 **Change atom (absolute address)**

```
#2:N : change A#2 C to N

  ?6,4\,|,?6,#2:N

** #n : (1<=n<=3095)
```




No.18 **Change atom (relative address)**

```
-2:N : change A(-2) C to N

  ?6,4\,?6,-2:N

** -n : (1<=n<=999)
```




No.19 **Fuse ring (attached 1 bond)**

```
?6,3=?5 : fuse ?5 at B3

  ?6,3=?5

** Bn(n:-999<=n<=4095): bond number
```




No.20 **Fuse multi ring (attached 1 bond)**

```
?6,$(-3,-4,-4,-2,-2,-4,-4)?6
?6,$(4,8,13,20,25,28,33)?6
```



No.21 **Fuse ring (attached 2 bond)**

```
(4,11)=?6[4]  : fuse 4/6 ring to B11..B4
(4,11)=?5[3]  : fuse 3/5 ring to B11..B4
(4,11)=?4[2]  : fuse 2/4 ring to B11..B4

MCd(1,.7)(  0,0)(<30,?6,3=?6,(11,4)=?6[4])
MCd(1,.6)(.54,1)(<30,?6,3=?6,(11,4)=?5[3])
MCd(1,.6)(  1,0)(<30,?6,3=?6,(11,4)=?4[2])
```



No.22 **Fuse ring (attached 3,4 bond)**

```
(16,4)=?6[3] : fuse 3/6 ring to B16..B4
(16,4)=?5[2] : fuse 2/5 ring to B16..B4
(21,4)=?6[2] : fuse 2/6 ring to B21..B4

MCd(1,.55)( 0,0)(?6,$(3,10)?6,(16,4)=?6[3])
MCd(1,.55)(.5,1)(?6,$(3,10)?6,(16,4)=?5[2])
MCd(1,.68)(1, 0)(?6,$(3,10,15)?6,(21,4)=?6[2])
```



No.23 **Spiro ring**

```
4*,?5 : add ?5(5 membered ring) at A4

<30,!6,4*,?5

An* : jump to An
```



No.24 **Branch bond (1)**

```
2\  : 2*,0
4*\ : 4*,0~wf
6\* : 6*,0~zf
8\\ : 8*,0~dm

MCf(<30,!8,2\,!,4*\,!,6\*,!,8\\,!)
```



No.25 **Branch bond (2)**

```
2\~dr  : 2*,0~dr
4\‘1.5 : 4*,0‘1.5
6\^15  : 6*,0^15

MCf(<-30,!6,2\~dr,!,4\‘1.5,!,6\^15,-60)
```



No.26 **Insert substituent(1)**

```
MCf(<30,
!,/Me,!,/Et,!3,/Pr,!,/iPr,!3,/tBu,!,/Ph^-30,!)

** Me:methyl Et:ethyl Pr:propyl iPr:isopropyl
   tBu:tertial buthyl Ph:phenyl
```



No.27 **Insert substituent(2)**

```
/  : single
// : double
*/ : wedge forward
/* : wedge dotted forward
** : direct

MCf(<30,!,//O,!,/*H,!,*/H,!,/?3,!,**?3,!)
```




No.28 **Insert substituent(3)**

```
~,^,‘,> : change type,angle,length,enviroment
          of substituent

MCf(<30,‘‘1,
    !,/Me~zf‘2^30,!,/Me~zf‘2^30,
    !2,*/Pr>lr,!2,*/Pr>rl,!)
```



No.29 **Add substituent(1)**

```
sw_numberA:=1; numberA_end:=12;
MCf(<30,!11,
  2:/Me,3:/Et,6:/Pr,7:/iPr,
  10:/tBu,11:/Ph^-30)
```




No.30 **Add substituent(2)**

```
~,^,‘ : change type,angle,length
        of substituent

MCf(<30,?6,
    @(3‘2^30,3~wf,4~zf,6~wf^-30,6~zf^30)/H)
```



## No.31 Add substituent(3)

```
~,^,‘,> : change type,angle,length,
          environment of substituent

MCF(<30,!7‘1,
@(2,3)/*Me‘2^30,5:*/Pr>lr,7:*/Pr>rl)
```




## No.32 Chain strech direction environment (1)

```
>hz : horizontal enviroment (default)
>vt : vertical enviroment

  ?4,
  @(3^-90,3^-30,3^90)/’(!3,"{hz}")>hz,
  @(1^-60,1‘2,1^60)/’(!2,"{vt}")>vt
```




## No.33 Chain strech direction environment (2)

```
>lr : left-right enviroment
>rl : right-left enviroment

  <30,!6,
  @(3^-30,3,3^30)/’(!3,"{lr}")>lr,
  @(5^-30,5,5^30)/’(!3,"{rl}")>rl
```




## No.34 Chain strech direction environment (3)

```
 >45 : fixed angle enviroment
 >’(-90,90,-90) : multi angle enviroment

  <-30,!6,@(2>45)/’(!3,"{45}"),
  @(6>’(-90,90,-90))/’(!3,"{(-90,90,-90)}")
```




## No.35 Change atom and Substituent

```
NH,SO,SOO : inset hetero atom and substituent
            simultaneously

  <30,!2,NH,!,SO,!,SOO,!3
```



## No.36 Change color

```
@(5)green : change color of A5 green
$(3)red :   change color of B3 red

  <30,Ph,@(2,5)N,
          2:red,5:blue,3=green

**  METAFONT ignore color command
```




## No.37 Chain start multiple characters

```
if chain start multi charactor string,
use !0 instead of !

  MCd( 1,1)(0  ,0.9)(<30,COOH,!0,!3,COOH)

  MCd(.8,1)(0.3,0.1)(<30,COOH,!4,COOH)
```




## No.38 User definition

```
iBuOH : user defined substructure

 iBuOH:='(!,/Me,!,OH)
 MCf(<30,?6,@(4,6)/iBuOH)
```




## No.39 Inline definition

```
Insert user defined substructure

  <30,!3,/'(!,/Me,!,OH),!3
```




## No.40 Font size

```
  beginfont("EN:Caffeine")
    font_wd#:=30mm#;  %<==font width
    font_ht#:=20mm#;  %<==font height
    sw_font_frame:=1;
    MCf(<30,?6,-4=?5,$(3,8)dl,@(2,6,7,9)N,
       @(2,6,9)/Me,@(1,5)//O)
  endfont
```



### No.41 Max ratio bond/width length

```
  max_bond_width:=0.10;
  MCd(1,1)( 0, .5)(<30,Ph)
  max_bond_width:=0.15;
  MCd(1,1)(.33,.5)(<30,Ph)
  max_bond_width:=0.25;
  MCd(1,1)( 1, .5)(<30,Ph)

** default: max_bond_width=0.15
```



### No.42 Ratio thickness/bond length

```
ratio_thickness_bond:= 0.005;
MCd(1,.6)(0, .5)(<30,Ph)
ratio_thickness_bond:= 0.015;
MCd(1,.6)(.5,.5)(<30,Ph)
ratio_thickness_bond:= 0.030;
MCd(1,.6)(1, .5)(<30,Ph)

** default: ratio_thickness_bond=0.015
```



### No.43 Offset thickness of bond

```
beginfont() offset_thickness#:=0pt#;
MCd(1,.3)(0, .5)(<30,Ph) endfont
beginfont() offset_thickness#:=0.2pt#;
MCd(1,.3)(.5,.5)(<30,Ph) endfont
beginfont() offset_thickness#:=0.5pt#;
MCd(1,.3)(1, .5)(<30,Ph) endfont

** default: offset_thickness#=0.2pt#
```



### No.44 Ratio char/bond thickness

```
ratio_char_bond:=1.0;
MCd(1,.6)(0, .5)(<30,?6,5:O,2:NH)
ratio_char_bond:=1.5;
MCd(1,.6)(.5,.5)(<30,?6,5:O,2:NH)
ratio_char_bond:=2.0;
MCd(1,.6)(1, .5)(<30,?6,5:O,2:NH)

** default: ratio_char_bond=1.5
```




### No.45 Ratio bondgap/bond length

```
ratio_bondgap_bond:= 0.10;
MCd(1,.6)(0, .5)(<30,Ph)
ratio_bondgap_bond:= 0.15;
MCd(1,.6)(.5,.5)(<30,Ph)
ratio_bondgap_bond:= 0.20;
MCd(1,.6)(1, .5)(<30,Ph)

** default: ratio_bondgap_bond=0.15
```



### No.46 Offset of doublebond gap

```
beginfont() offset_bond_gap#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,Ph) endfont
beginfont() offset_bond_gap#:=0.3pt#;
MCd(1,.3)(.5,.5)(<30,Ph) endfont
beginfont() offset_bond_gap#:=1.0pt#;
MCd(1,.3)(1, .5)(<30,Ph) endfont

** default: offset_bond_gap#=0.3pt#
```



### No.47 Ratio atom/bond length

```
ratio_atom_bond:= 0.25;
MCd(1,.6)(0, .5)(<30,?6,@(2,5)O)
ratio_atom_bond:= 0.36;
MCd(1,.6)(.5,.5)(<30,?6,@(2,5)O)
ratio_atom_bond:= 0.45;
MCd(1,.6)(1, .5)(<30,?6,@(2,5)O)

** default: ratio_atom_bond=0.36
```



### No.48 Offset of atom width

```
beginfont() offset_atom#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,Ph,@(2,4,6)N) endfont
beginfont() offset_atom#:=0.8pt#;
MCd(1,.3)(.5,.5)(<30,Ph,@(2,4,6)N) endfont
beginfont() offset_atom#:=2.0pt#;
MCd(1,.3)(1, .5)(<30,Ph,@(2,4,6)N) endfont

** default: offset_atom#=0.8pt#
```



### No.49 Ratio wedge/bond length

```
ratio_wedge_bond:=0.10;
MCd(1,.6)(0, .5)(<30,?6,5:*/Me)
ratio_wedge_bond:=0.12;
MCd(1,.6)(.5,.5)(<30,?6,5:*/Me)
ratio_wedge_bond:=0.20;;
MCd(1,.6)(1, .5)(<30,?6,5:*/Me)

** default: ratio_wedge_bond=0.12
```



### No.50 Offset of wedge width

```
beginfont() offset_wedge#:=0.0pt#;
MCd(1,.3)(0, .5)(<30,?6,5:*/Me) endfont
beginfont() offset_wedge#:=0.4pt#;
MCd(1,.3)(.5,.5)(<30,?6,5:*/Me) endfont
beginfont() offset_wedge#:=1.0pt#;
MCd(1,.3)(1, .5)(<30,?6,5:*/Me) endfont

** default: offset_wedge#=0.4pt#
```



No.51 **Ratio font atom gap/atom length**

```
ratio_atomgap_atom:=0.0;
MCd(1,.6)(0, .5)(<30,?6,@(2,5)O)
ratio_atomgap_atom:=0.050;
MCd(1,.6)(.5,.5)(<30,?6,@(2,5)O)
ratio_atomgap_atom:=0.12;
MCd(1,.6)(1, .5)(<30,?6,@(2,5)O)

** default: ratio_atomgap_atom=0.050
```



No.52 **Ratio chain/ring length**

```
ratio_chain_ring:= 0.4;
MCd(1,.6)(0, .5)(<30,?6,5:/Et)
ratio_chain_ring:= 0.66;
MCd(1,.6)(.5,.5)(<30,?6,5:/Et)
ratio_chain_ring:= 1.0;
MCd(1,.6)(1, .5)(<30,?6,5:/Et)

** default: ratio_chain_ring=0.66
```



No.53 **Ratio zebra gap/bond length**

```
ratio_zebragap_bond:=0.06;
MCd(1,.6)(0, .5)(<30,Ph,5:/*Me‘1)
ratio_zebragap_bond:=0.12;
MCd(1,.6)(.5,.5)(<30,Ph,5:/*Me‘1)
ratio_zebragap_bond:=0.20;
MCd(1,.6)(1, .5)(<30,Ph,5:/*Me‘1)

** default: ratio_zebragap_bond=0.12
```



No.54 **Margin left and right**

```
margin_left_right:=0mm;
MCd(1,1)(0.5,0.9)(<30,CH3,!0,!17,CH3)
margin_left_right:=0.4mm;
MCd(1,1)(0.5,0.5)(<30,CH3,!0,!17,CH3)
margin_left_right:=5mm;
MCd(1,1)(0.5,0.1)(<30,CH3,!0,!17,CH3)

** default: margin_left_right=0.4mm
```




No.55 **Margin top and bottom**

```
margin_top_bottom:=0mm;
MCd(1,1)(0.1,0.5)(<30,Ph,2:/OH,5:/NH2)
margin_top_bottom:=0.4mm;
MCd(1,1)(0.5,0.5)(<30,Ph,2:/OH,5:/NH2)
margin_top_bottom:=5mm;
MCd(1,1)(0.9,0.5)(<30,Ph,2:/OH,5:/NH2)

** default: margin_top_bottom=0.4mm
```



No.56 **Switch numbering atom**

```
numberA_start:=3; numberA_end:=8;

sw_numberA:=1; MCd(1,1)(.5,.9)(<30,!9)
sw_numberA:=2; MCd(1,1)(.5,.5)(<30,!9)
sw_numberA:=3; MCd(1,1)(.5,.1)(<30,!9)

** default: numberA_start=1 numberA_end=4095
```




No.57 **Switch numbering bond**

```
numberB_start:=3; numberB_end:=8;

sw_numberB:=1; MCd(1,1)(.5,.9)(<30,!9)
sw_numberB:=2; MCd(1,1)(.5,.5)(<30,!9)
sw_numberB:=3; MCd(1,1)(.5,.1)(<30,!9)

** default: numberB_start=1 numberB_end=4095
```




No.58 **Switch substituent off**

```
MCd(1,.5)( 0,0.5)(<30,Ph,4:/Cl,3:/F)

sw_subst_off:=1;
MCd(1,.5)( 1,0.5)(<30,Ph,4:/Cl,3:/F)

** default: sw_subst_off=0
```




No.59 **Switch all bond single**

```
MCd(1,.5)( 0,0.5)(<30,Ph,4:/Cl,3:/F)

sw_bond_single:=1;
MCd(1,.5)( 1,0.5)(<30,Ph,4:/Cl,3:/F)

** default: sw_bond_single=0
```




No.60 **Switch start vector**

```
sw_start_vector:=1;
MCf(<30,Ph,4:/Cl,3:/F)

** default: sw_start_vector=0
```



No.61 **Switch font frame**

```
sw_font_frame:=1;
MCf(<30,Ph,4:/Cl,3:/F)

** default: sw_font_frame=0
```




No.62 **Switch molecular frame**

```
MCd(1,.5)(0,0.5)(<30,Ph,4:/Cl,3:/F)

sw_mol_frame:=1;
MCd(1,.5)(1,0.5)(<30,Ph,4:/Cl,3:/F)

** default: sw_mol_frame=0
```




No.63 **Switch atom frame**

```
sw_atom_frame:=1;
MCf(<30,Ph,4:/COOH,3:/OH)

** default: sw_atom_frame=0
```




No.64 **Switch solid mode**

```
MCd(1,.8)( 0,0.5)(<30,Ph,4:/Cl,3:/F)

sw_solid:=1;
ratio_bond_width:=0.08;
MCd(1,.8)( 1,0.5)(<30,Ph,4:/Cl,3:/F)

** default: sw_solid=0
```




No.65 **Switch Expand**

```
MCd(1,.5)(0,0.5)(<30,Ph,4:/COOH,3:/NH2)

sw_expand:=1;
MCd(1,.5)(1,0.5)(<30,Ph,4:/COOH,3:/NH2)

** default: sw_expand=0
```



## No.66 Function MCd (draw)

```
MCd(a,b)(c,d)(...)
  a: ratio molecular width/font width
  b: ratio molecular hight/font hight
  c: x axis position
  d: y axis position

MCd(1,0.8)(0.2,0.9)(<30,Ph,2:/OH,5:/NH2)
MCd(1,0.8)(0.8,0.1)(<30,Ph,2:/OH,5:/NH2)
```




## No.67 Function MCf (fit draw)

```
MCf(...) : MCd(1,1)(0.5,0.5)(...)

sw_font_frame:=1;
sw_mol_frame:=1;
MCf(<30,Ph,2:/OH,5:/NH2)
```




## No.68 Local setting

```
beginfont()
  sw_mol_frame:=1; % <<== Local setting
  MCd(1,.4)( 0,.5)(Ph)
endfont
beginfont() MCd(1,.4)(.5,.5)(Ph) endfont
beginfont() MCd(1,.4)( 1,.5)(Ph) endfont
```



## No.69 Global setting

```
sw_mol_frame:=1; % <<== Global setting
beginfont() MCd(1,.4)( 0,.5)(Ph) endfont
beginfont() MCd(1,.4)(.5,.5)(Ph) endfont
beginfont() MCd(1,.4)( 1,.5)(Ph) endfont
```



## No.70 Output molecular information

```
var3:="calc_weight";  tag3:="cMW";
var4:="calc_formula"; tag4:="cFM";
%% Output to mcf_man_soc-info.aux %%
F:mcf_man_soc;C:85;cMW:194.19174;cFM:C8H10N4O2

cMW:calculated molecular weight
cFM:calculated molecular formula
```



### No.71 Output additional information

```
beginfont("EN:Tocopherol","CAS:59-02-9")
MCf(...) endfont

%% Output to mcf_man_soc-info.aux %%
F:mcf_man_soc;C:86;EN:Tocopherol;CAS:59-02-9

*F:filename *C:char number EN:molecular name
CAS:CAS number *:default output
```




### No.72 Change aux information delimiter

```
aux_delimiter:="/";
beginfont("EN:Warfarin","CAS:81-81-2")
MCf(...) endfont

%% Output to mcf_man_soc-info.aux %%
F:mcf_man_soc/C:87/EN:Warfarin/CAS:81-81-2

**default  aux_delimiter=";"
```




### No.73 Fixed aux information

```
tag3:="NO"; var3:="inf_NO";
tag4:="EN"; var4:="inf_EN";
sw_auxfix:=1; auxtag_out;
beginfont("EN:Ampicillin","NO:1") ... endfont

%% Output to mcf_man_soc-info.aux %%
F;C;NO;EN
mcf_man_soc;88;1;Ampicillin
```




### No.74 Example(1) Cholesterol

```
<30,?6,$(-4,-2)?6,-4=?5,7=dl,
1:*/OH,@(4,12)*/Me^60,9:*/H^60,
10:/*H^180,@(11,-1)/*H^-60,
-1\^17,/*Me,!4,/Me,!,
```




### No.75 Example(2) Colchicine

```
<30,Ph,@(1,2,6)/OMe,|,-4=?7,
|,-5=?7,$(-1,-4,-6)dl,-2://O,-3:/OMe,
#9\,NH,!,//O,!
```



No.76 **Example(3) Lycorine**

```
<30,Ph,-4=?6,-2=?6,6=?5,(9,12)=?5[3],
13=dl,8:N,@(15,17)O,
9:/*H^180,10:*/H^60,13:*/OH,14:/*OH
```



No.77 **Example(4) Gibberellin**

```
<12,?6'1.3,3=?5,9=?7,12\^160'1.6,&8,
4\^155~zf'1.2,O,55,//O^180'1,&2~zb,
5=dl,11=wf,13=wb,
7:/COOH,11://Me,1:*/OH,12:/*OH,2:*/Me,
@(3^-60,9^60)*/H
```



No.78 **Example(5) Quinine**

```
<30,Ph,3=Ph,7:N,6:/OMe,
 10\,*/OH,/H~zf^-60,!,
 |,?6,2:N,1:*/H^60,
 4*\,!~dr,
 2*,165~zf,60,&5~zb
```



No.79 **Example(6) Atoropin**

```
<-30,O,!,//O,!,!,Ph,
#1\~zb^-120,
|,?7'1.1,6*\^190'1.25,N,/Me,&3~wb,
#3\~wv,!,OH
```



No.80 **Example(7) Paclitaxel**

```
?6,5=d,3*,{,''1,36,45,45,45,45,},&#5,
-4=?6,-4=?4,-1=wb,-3=wf,-1:O,||,
@(4^35,4^-35,6)/Me,@(3^-60,15)*/OH,
8:/*H^-60,9:*/Me^60,10://O,
1\,O,!,//O,!,*/OH,!,/Ph,
60~wf,NH,-60,//O,60,Ph,
7\*,O,-45,//O,60,Ph,11*\,O,-60,//O,60,
12\*^-15,O,60,//O,-60
```